

10 Infinite Series

Have you ever wondered how a calculator or a computer can approximate the values of a function? In this chapter it is shown how to approximate functions by polynomial functions (including trigonometric polynomial functions). Just as irrational numbers can be approximated by rational numbers, which can be evaluated by “hand,” we will see that transcendental and rational functions can be approximated (locally) by polynomials. This is an improvement to the idea of approximating (locally) a differentiable function by its tangent line (which can be viewed as given by a polynomial of degree one). You will also see how trigonometric polynomials can be used to approximate functions over an entire interval.

10.1 Taylor Polynomials

In Chapter 2 it was shown that a differentiable function can be approximated by its tangent line in the neighborhood of a point.

Tangent Line Approximation at a Point

Let $f(x)$ be defined and differentiable in a neighborhood of the point $x = a$. Then for x near a the function $f(x)$ is approximated by

$$f(a) + f'(a)(x - a),$$

i.e., we write

$$f(x) \approx f(a) + f'(a)(x - a).$$

Example 10.1.1 Find the linear approximation of $f(x) = \sin x$ near $x = 0$.

Solution: The slope of the tangent line at $x = 0$ is

$$D(\sin)(0) = \cos 0 = 1.$$

The equation of the tangent line at $x = 0$ is

$$y = x,$$

and, hence, for x near 0

$$\sin x \approx x.$$

The following Maple V segment calculates the value of \sin at the points $-0.1, -0.05, -0.01, .01, 0.05$, and 0.1 .


```
> map(sin, [-0.1, -0.05, -0.01, 0.01, .05, 0.1]);
[-.09983341665, -.04997916927, -.009999833334, .009999833334, .04997916927,
.09983341665]
```

It is evident from the last Maple V output that $\sin x$ is well approximated by these values of x . Indeed the absolute values of the errors in the approximation are computed as follows.

```
> map(x -> abs(sin(x)-x), [-0.1, -0.05, -0.01, 0.01, .05, 0.1]);
[.00016658335, .00002083073, .166666*10-6, .166666*10-6, .00002083073,
.00016658335]
```

The fact that x is a good approximation to $\sin x$ only locally is illustrated in Figure 81.

```
> plot({sin(x), x}, x=-Pi/2..Pi/2);
```

Figure 81: Plot of $\sin x$ and x on $[-\pi/2, \pi/2]$

In general, a polynomial of degree n , whose derivatives at some point agree with the first n derivatives of $f(x)$, at that point is called a *Taylor Polynomial of $f(x)$ of degree n* at the point.

Taylor Polynomial Approximation of Degree n at a Point

Let $f(x)$ be defined and n times differentiable in a neighborhood of the point $x = a$. Then for x near a the function $f(x)$ is approximated by

$$f(a) + f'(a)(x-a) + \frac{f''(a)}{2!}(x-a)^2 + \cdots + \frac{f^{(n)}(a)}{n!}(x-a)^n,$$

i.e., we write

$$f(x) \approx f(a) + f'(a)(x-a) + \frac{f''(a)}{2!}(x-a)^2 \cdots \frac{f^{(n)}(a)}{n!}(x-a)^n.$$

When a function has an infinite number of derivatives in the neighborhood of $x = a$, then it has a Taylor Polynomial Approximation of Degree n for every positive integer n . In this case, we speak of a *Taylor's Series*.

Taylor Series at a Point

Let $f(x)$ be defined and have infinitely many derivatives in a neighborhood of the point $x = a$. Then the series

$$f(a) + f'(a)(x-a) + \frac{f''(a)}{2!}(x-a)^2 + \cdots + \frac{f^{(n)}(a)}{n!}(x-a)^n + \cdots$$

is called the *Taylor Series Expansion of $f(x)$ at $x = a$* .

When $a = 0$ the series is called a *Maclaurin Series*.

The Maple V procedure **taylor** provides a method for obtaining the Taylor Series expansion up through any value of n . The syntax for using this procedure is **taylor**(*expr*, *eq/nm*, n), where *expr* is an expression *eq/nm* is an equation (such as $x = a$) or name (such as x) and n is a (optional) non-negative integer.

For example, we can apply **taylor** to the \sin function at $x = 0$. (Since $a = 0$ here we are actually finding the Maclaurin Series.)

```
> f := taylor(sin(x), x=0);
```

$$f := x - \frac{1}{6}x^3 + \frac{1}{120}x^5 + O(x^6)$$

The term $O(x^6)$ indicates that the terms that are not shown in the series all contain a factor of x^6 . The default value of n is 6. If you want the series to more or less terms through order 5 then you should elect to set the option for n to a different value. Thus if you wish to obtain the Taylor Series for the sin function at $x = 0$ through order 9 you can issue the following:

```
> taylor(sin(x), x=0, 10);
```

$$x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7 + \frac{1}{362880}x^9 + O(x^{10})$$

The next Maple V segment indicates that the derivatives of the Taylor Series and the function agree at the point $x = a$.

```
> seq(subs(x=0, diff(f, x$i)), i=1..5);
```

1, 0, -1, 0, 1

```
> seq(eval(subs(x=0, diff(sin(x), x$i))), i=1..5);
```

1, 0, -1, 0, 1

You may use the **convert** procedure to convert a Taylor Series expansion to a Taylor Polynomial.

```
> convert(f, polynom);
```

$$x - \frac{x^3}{6} + \frac{x^5}{120}$$

You can use this whenever you want to use the Taylor Polynomial to approximate the function by the polynomial or to plot the approximation.

The following Maple V segment creates a list of the Taylor Polynomials for sin at $x = 0$ of degrees 1 through 5. Note that, for this function, the polynomials of even degree are actually odd degree polynomials of one less degree since all even coefficients are zero.

```
> P := [seq(convert(taylor(sin(x), x=0, i), polynom), i=2..6)];
```

$$P := [x, x - \frac{x^3}{6}, x - \frac{x^3}{6}, x - \frac{x^3}{6} + \frac{x^5}{120}]$$

In Figure 82 plot the Taylor Polynomials of degree 1, 3, and 5 along with the function sin over the interval $[-\pi, \pi]$.

```
> plot({sin(x), P[1], P[3], P[5]}, x=-Pi..Pi);
```

You should be able to identify the various Taylor Polynomials of sin in Figure 82 and observe that as the degree increases the approximation gets better.


You can also use **taylor** to obtain the Taylor Polynomials of arbitrary degree about any point, say, for example, setting $x = 1$ and $n = 3$ gives a Taylor Polynomial of degree 2 at $x = 1$.

```
> taylor(sin(x), x=1, 3);
```

$$(\sin(1) + \cos(1)(x-1) - \frac{\sin(1)}{2}(x-1)^2 + O((x-1)^3))$$

The

```
> convert(" ", polynom);
```

Figure 82: Plot of $\sin x$ and several Taylor Polynomials on $[-\pi/2, \pi/2]$

$$\sin(1) + \cos(1)(x-1) - \frac{\sin(1)(x-1)^2}{2}$$

Is there an analytic way of estimating the error made in approximating a function by a Taylor Polynomial? The following gives a way to bound the error.

Error Bound for Taylor Polynomial Approximations

Let $f(x)$ be defined and have $n+1$ derivatives in a neighborhood of the point $x=a$. Define the error term $E_n(x)$ by

$$E_n(x) = f(x) - (f(a) + f'(a)(x-a) + \frac{f''(a)}{2!}(x-a)^2 + \cdots + \frac{f^{(n)}(a)}{n!}(x-a)^n).$$

Suppose that $|f^{(n+1)}(x)| \leq M$ for $|x-a| \leq d$, then

$$|E_n(x)| \leq \frac{M}{(n+1)!} |x-a|^{n+1}$$

for $|x-a| \leq d$.

Example 10.1.2 Give a bound on the error made in approximating xe^{-x} by its tenth degree Taylor Polynomial about $x=0$, over the interval $-0.75 \leq x \leq 0.75$.

Solution let $f(x) = xe^{-x}$. Then the maximum value, M , of the eleventh derivative function of $f(x)$ must be found. Define $f(x)$, in a Maple V session and then find the eleventh derivative.

```
> f := x -> x*exp(-x);
```

$$f := x \mapsto xe^{-x}$$

```
> f11 := diff(f(x), x$11);
```

$$f_{11} := 11e^{-x} - xe^{-x}$$

In Calculus I the problem of finding a maximum value of a function on a closed interval was studied. One of the better ideas for solving such a problem is to plot its graph. See Figure 83.

```
> plot(f11, x=-0.75..0.75);
```




Figure 83: Plot of the eleventh derivative of xe^{-x} .




Figure 84: Plot of twelfth derivative of xe^{-x} .

Figure 84 suggests that the eleventh derivative is decreasing on the interval $[-0.75, 0.75]$. If this is true the maximum value of the eleventh derivative is equal to its value at the left-hand end-point $x = -0.75$. We investigate the sign of the twelfth derivative of $f(x)$.

```
> f12 := diff(f(x), x$12);
```

```
> plot(f12, x=-0.75..0.75, y=-30..0);
```

See Figure 84 for evidence that the twelfth derivative is always negative and hence the maximum value M of the eleventh derivative is obtained as follows.

```
> M := evalf(subs(x=-0.75, f11));
```

$$M := 24.87475020$$

Now we can obtain a bound on the error made by approximation $f(x) = xe^{-x}$ by its tenth order Taylor Polynomial with about $x = 0$. The value that d , given in the boxed statement, assumes in this example is $d = 0.75$. The bound is thus

```
> evalf(M*((0.75)^11)/(11!));
```

$$.2631945594 \cdot 10^{-7}$$

It follows that the error in approximating xe^{-x} by

```
> convert(taylor(f(x), x=0, 11), polynomial);
```

$$x - x^2 + \frac{x^3}{2} - \frac{x^4}{6} + \frac{x^5}{24} - \frac{x^6}{120} + \frac{x^7}{720} - \frac{x^8}{5040} + \frac{x^9}{40320} - \frac{x^{10}}{362880}$$

is bounded by

$$.2631945594 \cdot 10^{-7}.$$

Exercises 10.1

1. Find the Maclaurin Series expansion for $\cos(3x^2 + x)$ though terms of order 10.
2. Use the Maclaurin Series expansion of $f(x) = \cos(3x^2 + 2)$ to compute the values of the third, fourth, and fifth derivatives of $f(x)$ at $x = 0$. Use **coeff**, but do not use **diff**.
3. Determine the maximum error that could occur by using a Taylor Polynomial of degree 5 expanded about $x = 0$ to approximate $f(x) = \cos(2x) \sin(x)$, in the interval $-1 \leq x \leq 1$.

10.2 Trigonometric Polynomials

In the preceding section the problem of approximating a function in the neighborhood of a point by Taylor Polynomials was studied. These approximations are local in the sense that they (typically) are accurate only within some neighborhood. In this section trigonometric polynomials are utilized to approximate functions throughout a fixed interval, which is not assumed to be small.

The Fourier Series for $f(x)$ on $[-\pi, \pi]$

Let $f(x)$ be defined and integrable on $[-\pi, \pi]$, then $f(x)$ has a *Fourier Series Expansion* of the form

$$a_0 + a_1 \cos x + a_2 \cos 2x + a_3 \cos 3x \cdots \\ b_1 \sin x + b_2 \sin 2x + b_3 \sin 3x \cdots,$$

where

$$a_0 = \frac{1}{2\pi} \int_{-\pi}^{\pi} f(x) dx, \\ a_k = \frac{1}{\pi} \int_{-\pi}^{\pi} f(x) \cos kx dx, \quad \text{for } k > 0, \\ b_k = \frac{1}{\pi} \int_{-\pi}^{\pi} f(x) \sin kx dx, \quad \text{for } k > 0$$

An infinite series of the form

$$a_0 + \sum_{k=1}^{\infty} a_k \cos kx + \sum_{m=1}^{\infty} b_m \sin mx,$$

is called the *Fourier Series Expansion of $f(x)$* , and a partial sum, $S_n(x)$ of

$$S_n(x) = a_0 + \sum_{k=1}^n a_k \cos kx + \sum_{m=1}^n b_m \sin mx,$$


is called the *Fourier Polynomial of degree n of $f(x)$* .

Example 10.2.1 Find Fourier Polynomials for the function

$$f(x) = \begin{cases} 0 & \text{if } -\pi \leq x < 0, \\ 1 & \text{if } 0 \leq x < \pi \end{cases}$$

Make some maple V plots that compare these polynomials with the function. **Solution:** This function is nothing more than the Heaviside Function, H , restricted to the interval $[-\pi, \pi]$. See Chapter 1.8. The following Maple V makes defines $f(x)$ and makes a plot. See Figure 85.

```
> alias(H=Heaviside);
                                     I, H
> plot(H(x), x=-Pi..Pi, y=0..1.2);
```

Figure 85: Plot of the Heaviside Function on $[-\pi, \pi]$

The first component a_0 is the average value of $f(x)$ on the interval $[-\pi, \pi]$. Therefore, it is clear without the necessity of calculating an integral that

$$a_0 = 1/2.$$

For $k > 0$, $a_k = 0$, as can be seen from

$$a_k = \frac{1}{\pi} \int_{-\pi}^{\pi} f(x) \cos kx \, dx = \frac{1}{\pi} \int_0^{\pi} \cos kx \, dx = 0.$$

The b_k are obtained by

$$b_k = \frac{1}{\pi} \int_{-\pi}^{\pi} f(x) \sin kx \, dx = \frac{1}{\pi} \int_0^{\pi} \sin kx \, dx = \frac{1}{\pi} \cdot \frac{1 - \cos k\pi}{k}.$$

This simplifies to

$$b_k = \begin{cases} \frac{2}{k\pi} & \text{if } k \text{ is odd,} \\ 0 & \text{if } k \text{ is even} \end{cases}$$

Thus the Fourier Expansion of order n of $f(x)$ is given by

$$S_n(x) = 1/2 + \sum_{k=1}^m \frac{2}{2k-1} \sin(2k-1)x,$$

where $n = 2m - 1$. The following Maple V segment makes these computations and prepares Maple V plots. Since we already know that $a_0 = 1/2$, and if $k > 0$, then $a_k = 0$, we compute only the b_k .

```
> b := k -> int(sin(k*x), x=0..Pi)/(Pi);
      b := k -> (-cos(k*Pi)/k + k^-1) * Pi^-1
```

The following evaluates b_k for $k = 1, \dots, 4$.

```
> seq(b(k), k=1..4);
```

$$\left[\frac{2}{\pi}, 0, \frac{2}{3\pi}, 0\right]$$

Define the Fourier Polynomial of Order n as follows.

```
> k := 'k': S := n -> 1/2 + sum(b(k)*sin(k*x), k=1..n);
```

$$S \rightarrow 1/2 + \sum_{k=1}^n b_k \sin kx$$

The following calculates the Fourier Polynomials of orders 1 and 3.

```
> k := 'k': SS := [seq(S(2*n-1), n=1..2)];
```

$$SS := \left[1/2 + \frac{2 \sin(x)}{\pi}, 1/2 + \frac{2 \sin(x)}{\pi} + \frac{2 \sin(3x)}{3\pi}\right]$$

The next command produces a multiple plot of $f(x)$ along with the Fourier Polynomials of orders 1 and 3. See Figure 86.

```
> plot({H(x), SS[1], SS[2]}, x=-Pi..Pi, scaling=CONSTRAINED);
```




Figure 86: Multi-plot of Heaviside Function with Fourier Polynomials on $[-\pi, \pi]$

Figure 87 is a multiple plot of $f(x)$ along with the Fourier Polynomial of degree 9.

```
> plot({H(x), S(9)}, x=-Pi..Pi, scaling=CONSTRAINED);
```

Figure 87 indicates that the Heaviside Function is approximated by the Fourier Polynomial of degree 9, *i.e.*

$$H(x) \approx 1/2 + \frac{2 \sin(x)}{\pi} + \frac{2 \sin(3x)}{3\pi} \dots \frac{2 \sin(9x)}{9\pi}.$$

Exercises 10.2 Follow the example in this section to find and make plots comparing several Fourier Polynomials with each of the following functions.

1.

$$f(x) = \begin{cases} x + \pi & \text{if } -\pi \leq x < 0, \\ \pi - x & \text{if } 0 \leq x < \pi \end{cases}$$

2.

$$f(x) = \begin{cases} 0 & \text{if } -\pi \leq x < 0, \\ \sin x & \text{if } 0 \leq x < \pi \end{cases}$$




Figure 87: Multi-plot of Heaviside Function with Fourier Polynomial of degree 9 on $[-\pi, \pi]$