
              

9 Ordinary Differential Equations

Thoughout this book we have studied the two major concepts of calculus: the derivative, and the integral. In this
Chapter we begin the study of differential equations. Consider a problem that we can solve right now. Suppose
that we know the derivative of a function, x(t), and the value of the function at one value of time, for example,
suppose that it is given that

x′(t) = 2 t − 2

x(0) = 2.

Can we find this function? In terms of mathematics, this is the same problem as the one studied in Chapter 6, where
it was shown how to construct a function from a knowledge of its derivative. To find x(t) one asks what function
has a derivative equal to 2 t − 2? It is possible to find all such functions by taking the antiderivative. Thus x(t) has
the form x(t) = t2 − 2 t + C, where C is a constant. Since the function x(t) satisfies x(0) = 2, the constant C can
be found from

2 = 02 − 2 0 + C.

Hence a function x(t) which satisfies the two equations above is x(t) = t2 − 2 t + 2. Since any other function,
which satisfies the first equation must differ from x(t) by only a constant, and since it also satisfies the second
equation, it turns out that

x(t) = t2 − 2 t + 2

is the unique solution to the problem which was posed.
The problem of finding the antiderivative of a function is an example of a differential equation. Suppose that

f (t) is a continuous function defined over some interval [a, b]. A solution of the differential equation

x′(t) = f (t)

is an antiderivative of f (t),

x(t) =
∫

f (t) dt + C.

Since whenever x(t), defined in this way, is substituted for x into the equation x′ = f (t) equality holds, x(t) is
justifiably called a solution of the differential equation. In the next section we will define what a first order differ-
ential equation is and what is meant by a solution. The remainder of this chapter is concerned with some ways to
solve differential equations and how differential equations can be used to solve many problems that arise in several
fields of study.

9.1 What is a Differential Equation?

Water left in a glass cools or heats to the temperature of the surrounding air. If you drop an object into a body of
water, then the object eventually approaches the the temperature of the water. These observations are examples of
a general physical law called Newton’s Law of Cooling. For example, let T (t) be the temperature of the object at
time t and Ts be the surrounding temperature, then according to Newton’s Law of Cooling the rate of change of
the temperature satisfies

dT
dt

= −k(T − Ts),

where k is a constant that depends on the physical properties of the object. This is an example of a differential
equation, where T is the dependent variable and t is the independent variable.

Definition
Let f (t, x) be a function of two independent variables t and x. An equation of the form

x′ = f (t, x)

is called a first order differential equation. The variable x is the dependent variable and
the variable t is called the independent variable. A differentiable function φ(t) which is
defined on some interval I and such that

φ′(t) = f (t, φ(t))

is satisfied for all t ∈ I is called a solution of the differential equation.

For the differential equation that we derived from Newton’s Law of Cooling the dependent variable was T
instead of x and the function f in the definition is

f (t, T ) = −k(T − Ts).

In the introductory section to this Chapter we saw how to solve a special type of differential equation, i.e. one
in which the second variable x is absent in f. Consider a situation in which the variable t is absent, for example,
suppose f (t, x) = 2 − x, i.e., the differential equation is

x′ = 2 − x.

Observe that this differential equation has the same form as the one derived from Newton’s Law of Cooling, except
that in this case x is used in place of T, k = 1, and Ts = 2. It will be shown how to to find a solution to this
differential equation in Section 9.4. For the time being we will show how to determine if a given function is a
solution. Let

φ(t) = 2 − Ce−t,

where C is a constant. It will now be shown that φ(t) is a solution to the differential equation

x′ = 2 − x.

In order to show that φ(t) is a solution, it must be verified that

φ′(t) = 2 − φ(t).

The left-hand side of the preceeding equality can be found by differentiating φ(t) :

Left − hand side = φ′(t) = Ce−t.

The right-hand side is obtained by algebraic manipulation:

Right − hand side = 2 − φ(t) = 2 − (2 − Ce−t) = Ce−t.

It follows that
φ′(t) = 2 − φ(t)

and, thus,
φ(t) = 2 − Ce−t

is a solution to the first order differential equation

x′ = 2 − x.

Later in Section 9.4, you will learn how to solve this equation by hand, but now an illustration using Maple V
to solve the equation will be given. The first step is to define the differential equation in a Maple V session.

> deq := diff(x(t),t)=2-x(t);

deq := d
dt

x(t) = 2 − x(t)

Notice that when the equation was originally written the independent variable t was not shown explicitly. When
defining differential equations for use with Maple V, one must express them in terms of the independent variable
(t in this case), otherwise the Maple V dsolve command which is about to used will not treat things correctly. This
command will be applied and some of its many options will be explored throughout this chapter, but for now the
dsolve will be applied in its most basic form. Roughly speaking the dsolve command does for differential equations
what solve and fsolve does for algebraic equations. The basic syntax is dsolve( diffeqn, vars), where diffeqn is a
differential equation, and vars is the variables to be solved. In this case we have:

> dsolve(deq,x(t));

x(t) = 2 + e−t C1

How does this result compare with the solution that was given above? Notice that if C = − C1, then there is no
difference. Observe further that the answer is in the form of an equation. In order to manipulate the solution it is
better to write it in terms of an expression or a function.

> phi := rhs(");

φ := 2 + e−t C1

This defines φ as an expression. If you wish the solution to be expressed as a function, then use unapply.

> phi := unapply(phi,t);

φ := t 7→ 2 + e−t C1

You can verify that φ is a solution by having Maple V perform the same tasks that are done when checking it
by hand. First calculate the left-hand side of the differential equation with φ substituted for x.

> LeftHandSide := diff(phi(t),t);

Le f tHandSide := −e−t C1

Next substitute φ(t) for x(t) into the right-hand side of the equation.

> RightHandSide : = 2 - phi(t);

RightHandSide := −e−t C1 t

Since the left- and right-hand sides agree, you may conclude that the function φ is a solution.
The constant C in the solution above is like the constants of integration that were encountered when finding

antiderivatives. The only difference is that it enters the definition of φ as a multiplication factor; and the constants
obtained from antidifferentiation are additive constants.

What is the significance of the constant? If one knows the value of the solution at one point then, just as with an
antiderivative, one can determine the solution completely. For example suppose that it is required to find a solution
of the differential equation

x′ = 2 − x

which also satisfies the initial condition
x(0) = 3.

Then one can solve for the constant by solving the equation

3 = 2 − C e−0

for C. This is easily solved by hand and it may be concluded that C = −1. Consequently, the solution is

φ1(t) = 2 + e−t.

Using Maple V, proceed as follows. First solve for C1.

> C1 := solve(phi(0)=3,_C1);

C1 := 1

Then substitute the value into the solution that contains the arbitrary constant.

> phi1 := t -> subs(_C1=C1,phi(t));

phi1 := t 7→ 2 + e−t

You can plot your solution. See Figure 45

> plot(phi1(t),t=0..5);

Figure 45: A solution curve

Figure 46: Several solution curves

Once you are fortunate enough to have a formula for the solution, you can use it to analyze its algebraic, geo-
metric and numerical properties. For example, you can make multiple plots. See Figure 46 for plots of the solutions
for C1 = −2,−1, 0, 1, 2.

> plt := [seq(plot(subs(_C1=i,phi(t)),t=0..5),i=-2..2)]:
> plots[display](plt);

Figure 46 suggests that regardless of the value that is assigned C1, every solution approaches the same hori-
zontal asymptote x = 2.

> limit(phi(t),t=infinity);
2

It follows that every solution of the equation x′ = 2 − x approaches 2 asymptotically as t → ∞. Recall that
this differential equation is equivalent to one that arises from Newton’s Law of Cooling when the constant k = 1
and Ts = 2. Thus the fact that all solutions approach Ts = 2 asymptotically is consistent with the statement that an
object’s temperature cools or heats to its surroundings.

Could we have anticipated this result before solving the equation? Suppose that φ(t) is a solution of the dif-
ferential equation and φ(t) < 2. Then the slope of the curve x = φ(t) satisfies

φ′(t) = 2 − φ(t) > 0.

This means that so long as φ(t) < 2, that φ(t) is increasing. With the same reasoning it follows that φ(t) is de-
creasing whenever φ(t) > 2. Moreover, when φ(t) is near 2, the slope of the curve x = φ(t) is almost 0. This
suggests that, but certainly does not prove, that φ(t) approaches 2 as t gets large.

In the beginning of this section you were given a definition of a first order differential equation. You might
wonder about the term first order. The order of a differential equation is equal to the highest order derivative that
occurs in the equation.

Definition
Let f (t, x, y) be a function of three variables t, x, and y. An equation of the form

x′′ = f (t, x, x′)

is called a second order differential equation. A differentiable function φ(t) which is de-
fined on some interval I and such that

φ′′(t) = f (t, φ(t), φ′(t))

is satisfied for all t ∈ I is called a solution of the differential equation.

As an example of a second order differential equation that arises in physical problems, consider a mass, m, that
is attached to a spring. According to the physical law, known has Hooke’s Law, the amount of force required to
stretch or compress a spring is proportional to the length of the stretch or compression. If x denotes the displace-
ment and k is the proportionality constant then the force F is

F(x) = k x.

Another law from physics, Newton’s Second Law, states that

mx′′ = −kx.

If we divide both sides by m, then

x′′ = − k
m

x,

so that

f (t, x, y) = − k
m

x.

Because of Newton’s Second Law second order differential equations tend to come up in many problems involving
the motion of masses, and are thus of great interest.

As a particular example, of a second order differential equation consider

d2x
dt2

+ 2
dx
dt

+ 25 x = 0.

In this case

x′′ = d2x
dt2

and
f (t, x, y) = −2 y − 25 x.

Differential equations like this can arise in many problems, for example, is spring-mass systems with resistance.
It can be entered into a Maple V session as follows.

> eqn := diff(x(t),t$2)+2*diff(x(t),t)+25*x(t)=0;

eqn := d2

dt2
x(t) + 2

d
dt

x(t) + 25 x(t) = 0

It can be verified that
φ(t) = e−t cos(2

√
6 t)

is a solution of the differential equation by direct substitution and simplification.

> subs(x(t)=exp(-t)*cos(2*sqrt(6)*t),eqn);

d2

dt2
(e−t cos(2

√
6 t)) + 2

d
dt

(e−t cos(2
√

6 t)) + 25 (e−t cos(2
√

6 t)) = 0

> simplify(");

0 = 0

Which shows that φ(t) is a solution. If you still feel that Maple V is not a useful tool, you might try verifying
this by hand.

Exercises 9.1 In the following verify each φ(t) satisfies the indicated differential equation and plot φ(t).

1. Show that φ(t) = e3t is a solution of the differential equation

x′ = 3x.

2. Show that φ(t) = sin(6t) is a solution of the differential equation

x′′ + 36 x = 0.

3. Show that φ(t) = − 1
t−5 is a solution of the differential equation

x′ = x2.

4. Show that φ(t) = 100et/10

1+et/10 is a solution of the differential equation

x′ = 0.001x(100 − x).

9.2 Direction Fields

Let
x′ = f (t, x)

be a first order differential equation. At each point in the (t, x)-plane where f (t, x) is defined, the right-hand side
of the equation gives a value of the derivative,

φ′(t) = f (t, φ(t)),

of the solution through that point. This derivative can be thought of as a the slope of a line segment through that
point. The collection of all such line segments is called the direction field, (sometimes also called the slope field),
for the differential equation. Maple V has a procedure that produces a plot of a direction field. The procedure is part
of DEtools package and is called DEplot1. The syntax for using this procedure is DEplot1(deq,vars,trange,inits,xrange,options),
where deq is the right-hand side of the first-order differential equation, vars is list of the variables that are used,
trange is the range over which the independent variable ranges, inits is a set consisting of the initial conditions of
the solutions which are to be plotted. If the inits is omitted, then only the direction field is drawn and no solutions
are plotted. The variable xrange is the range over which the dependent variable ranges. If only solution curves are
required then the inits must be non-empty, and the option arrows should be set equal to NONE.

Example 9.2.1 Use the Maple V to obtain the direction field for the differential equation

x′ = 2 − x.

Then make a multiple plot of the solution curves to the differential equation which satisfy the five initial equations
x(0) = 0, x(0) = 1, x(0) = 2, x(0) = 3, and x(0) = 4. Finally, make a plot which is a composite of the preceding
plots.

Solution: Since the DEtools package is to be used, make the call using with. Then apply the procedure DEplot1
to obtain the direction field shown in Figure 47. Use the option arrows = LINE.

> with(DEtools):
> plt1 := DEplot1 ( 2 - x,[t,x],t=0..5,x=-4..4,arrows = LINE):";

Figure 47: Direction field for x′ = 2 − x

Figure 48: Five solution curves for x′ = 2 − x

Figure 49: Solution curves and direction field

Now we plot the five solution curves corresponding to the solutions which satisfy the initial conditions:

x(0) = 0, x(0) = 1, x(0) = 2, x(1) = 3, and x(2) = 4.

In the previous section we made similar plots and we could use the same method that was used in there, but DEplot1
will be used again. See Figure 48 and compare it with Figure 46.

> plt2 := DEplot1(2 - x,[t,x],t=0..5,{[0,0],[0,1],

> [0,2],[0,3],[0,4]}, arrows = NONE): ";

Figure 47 shows a collection of line segments through points in the (t, x)-plane. Each of these line segments is
tangent to the solution curve of x′ = 2 − x through that point. A good way to see this is to combine this plot with
some solution curves as those in Figure 48. See Figure 49.

> plots[display]({plt1,plt2});

Figure 48 can also be created by f a single call to the procedure DEplot1.

> DEplot1(diff(x(t),t)=2-x(t), [t,x],t=0..5,{[0,0],[0,1],[0,2],

> [0,3],[0,4]},arrows= LINE,x=-4..4);

Now consider a differential equation which comes up in population growth models called the logistic equation.
Let x(t) be the population of a certain species at time t. Assume a certain birth rate, kx, which causes the population
to grow. If there is nothing to check this growth, then rate of growth of x(t) satisfies the differential equation

x′ = kx.

Assume k > 0, since we are assuming a birth rate, as opposed to death rate. In this special case growth takes the
form

φ(t) = Cekt,

where C is a constant. These solutions tend to ∞ exponentially and soon would overpopulate the universe. More
realistically there are factors that limit growth as a population increases. Thus for logistic growth assume that x′

is also proportional to an expression of the form 1 − x
M . Thus it is assumed that x satisfies a differential equation

of the form
x′ = kx(1 − x

M
).

In the next example k = 1 and M = 10.

Example 9.2.2 Use the Maple V to obtain the direction field for the differential equation

x′ = x(10 − x)

10
.

Then make a multiple plot of the solution curves to the differential equation which satisfy the five initial equations
x(0) = 0, x(0) = 2, x(0) = 8, x(0) = 10, x(0) = 12 and x(0) = 20. Finally, make a plot which is a composite of
the preceding plots.

Solution We proceed as in the previous example. First call up the DEtools package. Then use DEplot1 to plot the
direction field. See Figure 50.

> with(DEtools):
> plt1 := DEplot1(x*(10 - x)/10,[t,x],t=0..5,x=-10..20,
> arrows = LINE):";

Note the line segments have positive, negative, or zero slope depending on where x is located. If φ(t) is a solution
of the equation and 0 < φ(t) < 10, then φ′(t) > 0. Thus in this range φ(t) is increasing. On the other hand if
φ(t) > 10 then φ′(t) < 0 and φ(t) is decreasing. Is the behavior of these solutions eseentially the same as those
of the previous example? At first glance you might feel that there is not much difference, but observe that for
φ(t) < 0 the slope is negative in Figure 50, but it is positive for Figure 47. The behavior is more complicated in
this example.

A plot of the solution curves satisfying the initial conditions:

x(0) = 0, x(0) = 2, x(0) = 8, x(0) = 10, x(0) = 12, and x(0) = 18

is given in Figure 51.

Figure 50: Direction field for x′ = x(10−x)

Figure 51: Five solution curves for x′ = x(10−x)

Figure 52: Solution curves and direction field

> plt2 := DEplot1(x*(10-x)/10,[t,x],t=0..5,{[0,0],[0,2],

> [0,8],[0,10],[0,12],[0,18]},x=-10..20,arrows=NONE): ";

Notice one difference with this example and the previous one is that the solution satisfying x(0) = 0, is the
constant function φ(t) = 0, and in the previous example all solutions tend to 2 as t tends to ∞. The union of the
last two plots is given in Figure 52.

> plots[display]({plt1,plt2});

The preceeding two examples do not contain t explicitly. The next example does.

Example 9.2.3 Use DEplot1 to analyze the time dependent system. Also duplicate the plots by using dsolve and
the plot command.

Solution: Starting as with the previous examples, we call up the DEtools package and invoke DEplot1 to create
Figure 53.

> with(DEtools):

> plt1 := DEplot1(t+x,[t,x],t=-4..4,x=-4..4,arrows = LINE): ";

Figure 53: Direction field for x′ = t − x

Figure 54: Two solution curves for x′ = t − x

A plot of solution curves satisfying the initial conditions x(0) = 0 and x(−1) = 0 is shown in Figure 54.

> plt2 := DEplot1(t+x,[t,x],t=-4..4,{[0,0],[-1,0]},

> x=-4..4,arrows=NONE): ";

It turns out that we can solve this equation with dsolve and then plot the curve over any scale that is available
to the plot command.

Define the equation in a Maple V session.

> deq := diff(x(t),t)=t+x(t);

deq := d
dt

x(t) = t + x(t)

Now assign the initial conditions and use dsolve.

> init1 := x(0)=0;
init1 := x(0) = 0

> sol1 := dsolve({deq,init1},x(t));

sol1 := x(t) = −t − 1 + et

In order to use plot it is necessary assign the right-hand side to an expression.

> x1 := rhs(sol1);

x1 := −t − 1 + et

Now plots are in Figure 55.

> plt2 := plot(x1,t=-4..4,x=-4..4):";

We now use the other initial condition and obtain Figure 56.

> init2 := x(-1)=0;
init2 := x(-1) = 0

> sol2 := dsolve({deq,init2},x(t));

Figure 55: A solution curve for x′ = t − x

Figure 56: Another solution curve for x′ = t − x

sol2 := x(t ) = - t - 1

> x2 := rhs(");

x2 := - t - 1

> plt3 := plot(x2,t=-4..4,x=-4..4):";

The direction field together with these last two curves is plotted in Figure 57.

> plots[display]({plt1,plt2,plt3});

Figure 57: Solution curves and direction field

Note that Figure 57 can be created in one step with DEplot1.

Exercises 9.2 Using Maple V procedures plot the direction fields for the given differential equations in the indicated
region of the (t, x)-plane . Include graphs of solution curves satisfying the indicated initial conditions.

1. x′ = −(t sin t) cos(x), where −2π ≤ t ≤ 2π, −2π ≤ x ≤ 2π. Initial conditions are x(−2) = 2, and x(−1) =
−3.

2. x′ = sin t cos t, where −2π ≤ t ≤ 2π, −2π ≤ x ≤ 2π. Initial conditions are x(−2) = 2, x(−2) = 3, and
x(0) = π.

3. x′ = cos(t − x), where −2π ≤ t ≤ 2π, and −2π ≤ x ≤ 2π. Initial conditions are x(0) = 1, x(0) = 0, and
x(0) = π.

4. x′ = t2 − x2, where −4 ≤ t ≤ 4, and −4 ≤ t ≤ 4. Initial conditions are x(0) = 0, x(−1) = −3, and x(1) = 2.

9.3 Euler’s Method

Numerical approximations of derivatives and integrals that were based on their definitions were presented in pre-
vious chapters prior to introducing techniques and shortcuts that can be calculating them in “nice” cases. In that
spirit, this section shows a method, (Euler’s Method), for approximating the solutions of initial value problems for
differential equations. The idea behind Euler’s Method is simple. Suppose that you wish to find the solution of

x′ = f (t, x)

which satisfies x(t0) = x0. Assume that f (t, x) varies continuously with its variables. Choose a small interval of
time, say h. Then since f (t, x) is continuous we can hope that f (t, x) is well approximated by f (t, x0), for |t − t0|
and |x − x0| small. One can then find the solution of the constant differential equation

x′ = f (t0, x0)

which satisfies x(t0) = x0. Integrating both sides gives the solution

φ(t) = x0 + f (t0, x0) (t − t0).

The solution at time t1 = t0 + h is
φ(t0 + h) = x0 + f (t0, x0) h

and so write
x1 = x0 + f (t0, x0) h.

Do the same thing again starting at the point (t1, x1) and obtain the approximate solution for the interval [t1, t1 + h]
to obtain

t2 = t1 + h, x2 = x1 + f (t1, x1)h.

After k steps it follows that
tk+1 = tk + h, xk+1 = xk + f (tk, xk)h.

This means that the approximate value of the solution to the initifal value problem at time t = tk is xk + f (tk, xk)h.

An illustration of this will now be given by approximating the solution of

x′ = t − x,

which satisfies x(0) = 1 using five iterations with h = 0.1

> f := (t,x) - > t - x;

f := (t, x) 7→ t − x

> t[0] := 0; h:=0.1; x[0] := 1;



            

9 ORDINARY DIFFERENTIAL EQUATIONS 92

t0 := 0

h := 0.1

x0 := 1

> t[1] := t[0]+h; x[1] := evalf(x[0] + f(t[0],x[0])*h);

t[1] := .1

x1 := .9

> t[2] := t[1]+h; x[2] := evalf(x[1] + f(t[1],x[1])*h);

t2 := .2

x2 := .82

> t[3] := t[2]+h; x[3] := evalf(x[2]+f(t[2],x[2])*h);

t3 := .3

x3 := .758

> t[4] := t[3]+h; x[4] := evalf(x[3]+f(t[3],x[3])*h);

t4 := .4

x4 := .7122

> t[5] := t[4]+h; x[5] := evalf(x[4]+f(t[4],x[4])*h);

t5 := .5

x5 := .68098

Thus the value of the solution to the initial value problem of the differential equation at times 0, 0.1, 0.2, 0.3, 0.4, 0.5
are 1, 0.9, 0.82, 0.758, 0.7122, 0.68098. In order to plot these points we create a list.

> L := [seq([t[i],x[i]],i=0..5)];

L := [[0, 1], [.1, .9], [.2, .82], [.3, .758], [.4, .7122], [.5, .68098]]

We can now make a plot of the approximate solution Figure 58.

> plot(L);

You can verify by direct substitution that the exact solution to this problem is

φ(t) = 1 − 2 e−t.

The error in the approximation at point ti is

ERROR = approximation at ti - φ(ti)

This can be computed at each of the points 0, 0.1, 0.2, 0.3, 0.4, 0.5.

> error := seq(evalf(L[i][2]-(-1+2*exp(-(i-1)/10))),i=1..5);

error := 0, .090325164, .182538494, .276363559, .371559908

You can compare the approximate and exact solution graphically as in Figure 59. For this problem the approximate
solution is the higher one.

> plot({L,rhs(sol)},t=0..0.5);

Figure 58: Euler’s Method solution for x′ = t − x

Figure 59: Approximate and exact solutions

When making a large number of iterations it is probably better to use a loop instead of typing in all of the
repetitions. The following maple V segment shows how to make 50 iterations with h = 0.01.

> t[0] := 0: h:=0.01: x[0] := 1:

> i := ’i’:

> for i from 0 to 49 do
> x[i+1] := x[i] + f(t[i],x[i])*h;
> t[i+1] := t[i]+h;

> od:

> x[50];

.7100121342

The new approximate value for t = 0.5 is x[50] in this case and is .7100121342 which is a better approximation
than the one with h = 0.1.

Exercises 9.3 Use Euler’s Method with step size equal to h = 0.1 to determine an approximate value of the solution
at t = 1 for each of the initial value problems below. Repeat these calculations with h = 0.05, and h = 0.01 and
compare the result with the exact value of x(1). You may use dsolve to obtain the exact solution. Graph the result
along with the direction field in each case.

1. x′ = x, x(0) = 1

2. x′ = t + x, x(0) = 1

3. x′ = t − x, x(0) = 2

4. x′ = 3x − 4e−t, x(0) = 1

5. x′ = x(10 − x), x(0) = 2

9.4 Separation of Variables

In the preceding sections you have seen how to use the direction field defined by a differential equation to gain
geometric insight into how the solutions behave, and how to use Euler’s Method to numerically approximate the
solutions. In this section a method for finding the exact solution, in cases when the differential equation is given
in the following special form:

x′ = T (t)X(x).

Definition
A first-order differential equation is separable if it can be written in the form:

x′ = T (t)X(x).

As an example consider the following differential equation:

x′ = (2t + 1)x.

This equation is in separable form with T (t) = 2t + 1, and X(x) = x. Suppose that φ(t) is a solution of x′ =
(2t + 1)x, then φ(t) satisfies

φ′(t) = (2t + 1)φ(t).

If φ(t) = 0 then the right hand side of the equation is zero which means that the constant function with value zero

φ(t) = 0

satisfies the equation. More generally, if φ(t) 6= 0 one can divide both sides of the equation by φ(t) obtaining

φ′(t)
φ(t)

= 2t + 1.

Observe that both sides of this last equation can be integrated with respect to t∫
φ′(t)
φ(t)

dt =
∫

(2t + 1) dt.

Calculating the integral on each side we obtain

ln φ(t) = t2 + t + C.

In order to solve for the solution φ(t) we apply the inverse function, exp, to both sides. Thus

φ(t) = et2+t+C.

This can be written as
φ(t) = Ket2+t,

where K = eC. The constant, K, can be evaluated by solving the last equation when t = 0, and a formula for φ is
found:

φ(t) = φ(0) et2+t.

Once you have found a candidate for a solution to a differential equation it is always a good idea to check to
see if it really satisfies the equation. Upon differentiating the equation for φ, with the help of the chain rule, you
arrive at

φ′(t) = φ(0)et2+t (2t + 1) = φ(t)(2t + 1) = (2t + 1)φ(t).

Thus φ(t) is a solution.

In general when an equation has the separable form you can obtain at least a implicit expression for a solution.
Suppose then that a differential equation of the form

x′ = T (t) X(x)

is given. How does one find the solution? In the example above it is assumed that x = φ(t) is a solution and then
both sides are divided by x = φ(t). Since in that example X(x) = x, divide the general separable equation by
X(φ(t)). Then

φ(t)′

X(φ(t))
= T (t).

Let G(x) denote the antiderivative of the function 1
X(x)

then if both sides of the last equation are integrated with
respect to t one obtains

G(φ(t)) =
∫

T (t) dt + C,

where the fact that

G(x) =
∫

dx
X(x)

,

implies

G(φ(t)) =
∫

φ′(t) dt
X(φ(t))

.

The equation

G(φ(t)) =
∫

T (t) dt + C

defines the function
φ(t)

implicitly. Since

G′(x) = 1
X(x)

6= 0,

the function G has an inverse, G−1, thus φ(t) is given by

φ(t) = G−1(

∫
T (t) dt + C).

Often one can’t find an elementary formula for G−1, but in any case the solution x = φ(t) is given implicitly by
the relation:

G(x) −
∫

T (t) dt = C.

A function F(t, x) like

F(t, x) = G(x) −
∫

T (t) dt

which is constant when a solution is substituted for x is called an integral for the differential equation. Thus for
the differential equation

x′ = (2t + 1)x

the function
F(t, x) = ln x − et2+t

is an integral for the differential under discussion.

Example 9.4.1 Use the method of separation of variables to find the solutions of the logistic type equation

x′ = x(x − 1).

Solution: This equation is in the separable form with T (t) = 1, and X(x) = x(x − 1). The process starts by di-
viding the equation by x(x − 1). With Maple V the session begins as follows.

> deq := diff(x(t),t) = x(t)*(x(t)-1);

deq := d
dt

x(t) = x(t) (x(t) − 1)

> deqsep := deq/(x(t)*(x(t)-1));

deq :=
d
dt x(t)

x(t) (x(t) − 1)
= 1

Now the variables are separated with the function of x on the left-hand side and the function of t on the right-hand
side. Each side can be integrated.

> intlhsdeqsep := int(lhs(deqsep),t);

intlhsdeqsep := − ln(x(t)) + ln(x(t) − 1)

> intrhsdeqsep := int(rhs(deqsep),t)+C;

intrhsdeqsep := t + C

This means that
F(t, x) = − ln(x) + ln(x − 1) − t

is an integral of the differential equation x′ = x(x − 1). Now solve the equation for x(t).

> phi := solve(intlhsdeqsep=intrhsdeqsep,x(t));

φ := 1(
1 − et+C

)
Sometimes it is desirable to make φ a function.

> phi := unapply(phi,t);

φ := t 7→ 1
1 − et+C

Thus the general solution is

φ(t) = 1
1 + et+C

.

The following steps represent a check to see if φ(t) really is a solution. The candidate for a solution is substituted
into the differential equation

> eval(subs(x = phi,deq));

et+C(
1 − et+C

)2 =
(
1 − eC

)−1 − 1

1 − et+C

Simplification illustrates the validity of the solution.

> simplify(");
et+C(−1 + et+C

)2 = et+C(−1 + et+C
)2

Since both sides are equal we have checked the correctness of the solution.
In this section the method of separation of variables is being emphasized. Now an illustration of how to solve

this equation using dsolve will be presented.

> dsolve(deq,x(t));

1
(x(t))

= 1 + et C1

Note that dsolve did not solve for the solution explicitly, but it is easy to obtain the explicit solution.

> solve(",x(t));

(x(t)) = 1
1 + et C1

This says that the general solution is

x(t) = 1
1 + et C1

.

You should be able to prove that this is equivalent to the one obtained above.
When you can find the solution explicitly, you can also use the Maple V option explicit to ask dsolve to return

that solution.

> dsolve(deq,x(t),explicit);

x(t) = − 1
(−1 − et C1 )

When a differential equation is separable you have a chance to find an exact solution. Sometimes it is not easy,
(or even possible), to evaluate the integral ∫

dx
X(x)

.

That can be an obstruction to finding an exact solution. At other times you can not obtain the solution explicitly.
Nevertheless the method of separation of variables appears in applications often enough to be studied. The next
example is an illustration of a problem in which Maple V’s ability to evaluate integrals helps a lot.

Example 9.4.2 Find the solution of

x′ = t2

ex cos(x)
√

9 − t2
,

which satisfies the initial condition x(0) = 0. Plot the graph of the solution.

Solution: This equation is separable with T (t) = t2√
9−t2

, and X(x) = ex cos x. Start out just like in the last problem.

> deq := diff(x(t),t)= (t^2)/(exp(x(t))*cos(x(t))*\sqrt(9-t^2));

deq := d
dt

x(t) = t2

ex(t) cos(x(t))
√

9 − t2

Now separate variables

> deqsep := deq * (exp(x(t))*cos(x(t)));

deqsep := ex(t) cos(x(t))
d
dt

x(t) = t2

√
9 − t2

and integrate both sides.

> intlhsdeqsep := int(lhs(deqsep),t);

intlhsdeqsep := ex(t) cos(x(t))
2

+ ex(t) sin(x(t))
2

> intrhsdeqsep := int(rhs(deqsep),t)+C;

intrhsdeqsep := − t
√

9 − t2

2
+ 9 arcsin( t

3 )

2
+ C

The integral follows by equating the last two results.

> integral := intlhsdeqsep = intrhsdeqsep;

integral := ex(t) cos(x(t))
2

+ ex(t) sin(x(t))
2

= − t
√

9 − t2

2
+ 9 arcsin( t

3 )

2
+ C

An explicit solution, in this case, seems impossible, so leave the solution in implicit form. Now solve for C by
using the initial condition.

> inits := eval(solve(subs({t=0,x=0},integral),C));

inits := 1/2

Substitute this value of C into the integral.

> initintegral := subs(C=1/2,integral);

initintegral := ex(t) cos(x(t))
2

+ ex(t) sin(x(t))
2

=

− t
√

9 − t2

2
+ 9 arcsin( t

3 )

2
+ 1/2

One can now plot the solution using implicitplot. See Figure 60.

> plots[implicitplot](initintegral,t=-3..3,x=-2*Pi..2*Pi);

Figure 60: Solution Curve

Exercises 9.4 Use the method of separation of variables to solve the following initial value problems. Whenever
possible find explicit solutions. In all cases plot the solution.

1. x′ = t(1 + x2), x(0) = 1

2. x′ = sin t sin x, x(0) = π
4

3. x′ = 0.005x(500 − x), x(0) = 20

4. x′ = 1
x ln x , x(0) = 1

5. x′ = t(1+x2)

(1+t2)
, x(0) = 1

9.5 Models of Growth and Decay: First Order Rates

Much of the work. that scientists and engineers do, involves the modeling process. In previous sections we have
made reference to physical laws such as, for example, Newton’s law of Cooling, or Hooke’s Law. These are exam-
ples of famous mathematical models and have been accepted by elements of the scientific community over such
a long time that they are called “laws”. Much of the routine daily activity of practicing scientists involves devel-
oping some kind of model. The process might start with some kind of “ real-world” problem, which comes up in
the scientist’s discipline, that may only be vaguely understood, but for which there are good reasons to have more
understanding. The goal to be achieved must be articulated. The process might go like this. In the first step one
determines components affecting the behavior of the problem under consideration, and isolates those mechanisms
that are important in terms of the overall goals. The problem is then cut down to a manageble size. The next step is
to determine constraints and scientific laws that apply to the specialized problem. Most mathematical models have
the following elements: (1) a mathematical or logical structure, (algebraic formulas, differential equations, etc.),
(2) definitions of the variables involved, and (3) the distinquishing features within the mathematical structure of all
laws and constraints that are relevent to the problem. Once a model has been obtained, it can be analyzed through
its own internal mathematical structure so that the behavior of its variables can be predicted. Thus the process that
starts with a “real-world” problem leads to a mathematical problem, which can be analyzed to obtain a prediction
about the original problem. The predictions made as a result of the model may or may not agree with experimental
results, or might suggest new laboratory experiments. If the model does not give realistic predictions then one must
return to the model, determine which assumptions made during the process have led to these incorrect predictions,
and then make revisions to the model accordingly.

Radioactive Decay

As a radioactive material loses some of its mass as radiation energy, the remainder of the material reforms to create
a new substance. This process is called radioactive decay. For example, as radioactive carbon-14 decays it forms
nitrogen. The ultimate result of the decay of radium is lead. Experiments have shown that at any given time, the
rate at which a radioactive element decays is proportional to the mass of the element that is present. Let x(t) denote
the mass of a radioactive substance at time t. Its rate of decay has the form

x′ = −kx.

If x0 is the mass at time t = 0, then
x(t) = x0e−kt.

The half-life of a radioactive substance is the time required for half of the substance to have decay. It is related to
k by solving for Thal f in the equation

x0e−kThal f = x0

2
.

Thus

Thal f = ln(2)

k
.

Example 9.5.1 A living substance is assumed to have the same proportion of carbon-14 as the atmosphere has
and stops absorbing carbon when it dies. This means that the proportion of carbon-14 in, say, a plant that was once
alive can be used as an indicator of how long ago the plant died. The half-life of carbon-14 is 5700 years. Suppose
that a sample has 90% of the carbon-14 that it originally had. Find the age of the sample.

Solution: Let x(t) denote the amount per gram of carbon-14 per gram of carbon in sample at time t. Then x(t) =
x0e−kt. Since the half-life of carbon-14 is 5700 years, we can determine k from the formula

k = ln(2)

5700
.

The time T in years back at which the sample died satisfies

x0e−kT = x0 · (9/10),

or
e−kT = 9/10.

Solving for T gives

T = −5700 ln(0.9)

ln(2)
.

> T = evalf(-5700*ln(0.9)/(ln(2)));

T = 866.4176331

It can be concluded that the sample has been dead for at least 866 years.

Drag Near the Earth’s Surface

A body of low density and rough exterior (e.g., a feather, or a snowflake), moving near the earth’s surface has a
resistive force due to air which is propotional to the velocity, v, but acts opposite to the motion. Thus if such a
body has mass m is released at height x0 with initial velocity v0 in the vertical direction has a force due to gravity
and resistance equal to

F = −mg − kv,

where g is the acceleration due to gravity, and k > 0 is a constant of proportionality. Using Newton’s Second Law
it can be seen that v(t) satisfies the initial value problem

mv′ = mg − kv, v(0) = v0.

Example 9.5.2 Suppose that the velocity v(t), of a body of low density satisfies the initial value problem

v′ = −100 − 0.04 · v, v(0) = 0.

Find the limiting velocity.

Solution: In this problem the differential equation is given so all that has to be done is to solve it.

> deq := diff(v(t),t) = -100- 0.4*v(t);

deq := d
dt

v(t) = −100 − 0.4v(t)

> dsolve({deq,v(0)=0},v(t));

Error, (in factor/factor) floats not handled

The error here occurs because when one is trying to find the exact solution using dsolve the differential equation
can not use floating point numbers. There are several ways to remedy this in this case. One way be to rewrite 0.4
as 4/10 = 2/5.

> deq := diff(v(t),t) = -100- (2/5)*v(t);

deq := d
dt

v(t) = −100 − 2 v(t)
5

The dsolve procedure works in this case and the exact solution can be found as follows.

> dsolve({deq,v(0)=0},v(t));

v(t) = −250 + 250 e− 2 t
5

The limiting velocity is -250 units of velocity. The significance of the negative sign is that it means that the object
is falling.

In other situations, such as, if the falling object is dense (e.g., a raindrop, baseball or bullet), and moves near
the earth’s surface, the resistive force of the air might be proportional to the square of the speed and acts opposite
to the direction of the motion. Therefore the equation of motion has the form

mv′ = −mg ± kv2,

where k > 0 is the drag coefficient; the upper sign (+) is chosen if the body is falling, and the lower sign (-) if the
body is rising. The equation can also be written

mv′ = −mg − v|v|.

Example 9.5.3 Suppose a smooth dense object falls with velocity which satisfies the differential equation:

v′ = −1 − v|v|
Find the limiting velocity in the case of an initial velocity of v(0) = 0, and plot several solution curves with various
initial velocities.

Solution: In this example dsolve with the numeric option will be used to solve the given initial value problem.

> deq := diff(v(t),t)=-1-v(t)* abs(v(t));

deq := d
dt

v(t) = −1 − v(t) |v(t)|

> sol := dsolve({deq,v(0)=0},v(t),type=numeric);

sol := proc(rkf45_x) ... end

We can estimate the limiting velocity.

> seq(sol(i)[2],i=3..5);

v(t) = -.9950558671716422, v(t) = -.9993294546617530,

v(t) = -.9999092277555853

It appears to be v = −1. The velocity tends to approach the constant solution v(t) = −1. Next use DEplot1 to plot
several solution curves. See Figure 61

> with(DEtools):

> DEplot1(deq,v(t),t=0..5,{[0,-2],[0,-1],[0,0],[0,1],[0,2],[0,3]},

> arrows = LINE,v=-2..3);

Population Models

Let P(t) denote population at time t of a species. In reality the values of P(t) are integers, and they change by
integral amounts with time. However, for large populations a change of one or two is “infinitessimal” relative to the
total, and we may think of P(t) as a continuous or even a smooth function, and thus we hypothesize the existence
of the rate of change of the population, P′(t). This leads to differential equations. In general the following relation
is assumed to hold:

Figure 61: Solution curves for velocity of smooth, dense falling object

Rate of Change = rate in - rate out

Example 9.5.4 Suppose that a population is isolated in the sense that there are no outside influences (i.e., there
is no immigration or or emigration) and that the only change in the population is due to births and deaths. In this
case the rate in is the birth rate and is assumed to be linearly proportional to the the size of the population

rate in = b· P(t)

The rate out is the death rate and is also assumed to be linearly proportional to to the size of the population.

rate out = d · P(t)

Then the size of the population P(t), with initial size, P(0), satisfies the initial value problem:

P′ = (b − d)P,

P(0) = P0.

The solution of this initial value problem can be obtained by finding the explicit solution using the method of sep-
aration of variables to be

P(t) = P0e(b−d)t.

If the birth rate exceeds the death rate then the population grows exponentially, but if the death rate is larger than
the population “dies out” exponentially.

Example 9.5.5 Suppose that a population initially has a birth rate constant of proportionality b = 0.06, births/year
and a death rate of d = 0.04 deaths/year. After 15 years of steady growth, assume that the population stops repro-
ducing, i.e, b = 0. Find how long after the population stops reproducing that it takes the population

1. to return to its original level

2. to reach 50% of its original level

3. to reach 30% of its population at the time b became zero.

Solution: In this case the differential equation changes with time. For the first fifteen years we have

P′ = 0.02 · P,

which has solution
P(t) = P0e0.02t.

After 15 years the differential equation is
P′ = −0.4 · P.

We can define the solution piecewise by
P(t) = P0e0.02t,

for 0 ≤ t ≤ 15, and
P(t) = P15e−0.4(t−15),

for 15 < t, where P15 = P0e.30. You need to verify this. Think of solving the initial value problem

P′ = −0.04P, P(15) = P0e0.02·15.

Using the Heaviside Function, H, you can define the solution

P(t) = P0e0.02t(H(t) − H(15) + P0e.30e−0.04(t−15) H(t − 15).

A plot of the solution curve along with the horizontal lines corresponding to the original population, 50% of
the original population, and 30% of the population at the time b becomes zero is shown in Figure 62 where we
have set P0 = 1 in order to plot the graph.

Figure 62: Population in which reproduction stops

In our Maple V session we use the notation Pbefore to denote the population before reproduction halts and
Pafter to denote the population after this event.

> Pbefore := (P0,t) -> P0*exp(0.02*t);

(P0 , t) 7→ P0 e0.02t

The population at the time reproduction seizes is found as follows.

> P15 := Pbefore(P0,15);

P15 := .349858808P0

Now compute Pafter.

> Pafter := (P0,t) -> Pbefore(P0,15)*exp(-0.04*(t-15));

Paf ter := (P0 , t) 7→ 1.349858808P0 e−0.04t+0.60

Use the Heaviside Function and plot Figure 62.

> alias(H=Heaviside);
I, H

> P := Pbefore(1,t)*(H(t)-H(t-15)) + Pafter(1,t)*H(t-15);

P := e0.02t (Heaviside(t) − Heaviside(t − 15)) + 1.349858808e−0.04t+0.60Heaviside(t − 15)

> plt1 := plot(P,t=0..100,0..1.5): ";

> plt2 := plot(1,t=0..100):

> plt3 := plot(1/2,t=0..100):

> plt4 := plot(subs(P0=1,0.30*P15),t=0..100):";

> plots[display]({plt1,plt2,plt3,plt4});

We now solve the problem numerically.

> solve(Pafter(P0,t) = P0,t);

22.50000001

> solve(Pafter(P0,t)=P0/2,t);

39.82867952

> solve(Pafter(P0,t)=0.30*P15,t);

45.09932010

Using the preceding Maple V segment one can now answer the questions posed in the example. You may
conclude that it takes

22.5 − 15 = 7.5

years for the population to return to its original level. The population returns to its original population in half of
the time that the population grew to its maximum value. The population reaches 50% of the original population in

39.82867952 − 145 = 24.82867952

years. Finally, the population decreases to 30% of its population at the time b became zero in

45.09932010 − 15 = 30.09932010

years.
The “explosive” growth that arises when a population satisfies a linear growth rate is not always realistic, since

the exponential increase will soon outstrip the resources that are necessary to support the population. One way to
model restricted population growth is to assume that the rate coeffient is variable rather than constant. Assume that
this coefficient is linear, the next simplest after the constant case, then the population can be assumed to satisfy the
logistic equation

P′ = r(L − P)P.

The factor L − P is called the limiting factor. This problem can be solved by the method of separation of variables.
There have been several versions of this type of equation analyzed in previous sections.

Example 9.5.6 It is known that the resources of a certain region can sustain at most 250 wolves. There are presently
25 wolves in the region. Assume that the population of wolves grows at a logistic rate and the constant of propor-
tionalty is r = 0.001 wolves/year.

1. Determine the population of wolves P(t) as an explicit function of time.

2. Plot the graph of P(t).

3. What values of P and t make sense in the problem situation?

4. When will the wolf population reach 100?

5. When will the population essentially reach its limit?

Solution:

1. The population of the wolves must satisfy the equation

P′ = .001P(250 − P),

with P0 = 25.

> P := ’P’;

> deq := diff(P(t),t) = 1/1000*P(t)*(250-P(t));

deq := d
dt

P(t) = P(t) (250 − P(t))
1000

Using dsolve with the explicit option, obtain the explicit solution and denote it by PW.

> PW := dsolve({deq,P(0)= 25},P(t),explicit);

PW := P(t) = − 250

−1 − 9 e− t
4

The expression PW is converted to a function using unapply.

> PW := unapply(rhs(PW),t);

PW := t 7→ − 250

−1 − 9 e− t
4

2. Now plot the graph of P(t). See Figure 63.

> plot(PW(t),t=-10..50,-50..300);

3. Since the population can never be negative, nor exceed 250, and since time is measured from the present, it
follows that:

0 ≤ t, 0 ≤ PW ≤ 250.

4. Use solve to predict when the population will reach 100.

> solve(PW(t)=100,t);
4 ln(6)

> evalf(");

7.167037876

Figure 63: Graph for P(t)

This indicates that there will be 100 wolves in about 7.167037876 years.

5. The wolf population will reach 249 (one less than the limit) in

> solve(PW(t)=249,t);
4 ln(2241)

> evalf(");

30.85870990

Therefore the population can be considered to reach its limiting population in around 31 years. See Figure 64.

> plot(PW(t),t=0..40,0..250);

Figure 64: Graph for wolf population



                

9 ORDINARY DIFFERENTIAL EQUATIONS 107

Equilibrium Solutions

Most of the differential equations that have been used as models have the form

x′ = f (x).

For example in the example on Newton’s Law of Cooling the differential equation had the form:

T ′ = −k(T − Ts).

In the equation for radioactive decay the differential had the form:

x′ = −kx.

The equation of population growth has the form

P′ = (b − d)P.

The equations for logistic population growth of wolves in the last example was

P′ = −.001P(250 − P).

In each case these equations have points where the right hand side of the equation vanishes. These are points in
which the derivative of the solution is zero. Since the solution curve through such a point has zero slope the solution
must be constant, i.e., is in equilibrium. Such points are called equilibrium points, and the constant solution is
called an equilibrium solution. How do solutions of a differential equation behave near an equilibrium point? In
the model involving Newton’s law of Cooling the equilibrium point is T = Ts, the temperature of the surrounding
area. All solutions tend to this temperature as t → ∞. One says that an equilibrium solution that has this property
is stable. In the population qrowth equation the equilibrium occurs at P = 0, the general is

P(t) = P0e(b−d)t.

If the birth rate is larger then the death rate then the growth “explodes” as t increases, but note that as t → −∞ the
solution tends to the equilibrium solution. When this happens it is said that the equilibrium solution is unstable.
On the other hand, if the death rate exceeds the birth rate all solutions tend to the equilibrium solution P = 0, i.e.,
the population dies out, i.e., the equilibrium solution is stable. Finally, in the logistic equation for wolves there are
two equilibrium points: P = 0 and P = 250. The solution of the initial value problem used for this equation was

P(t) = − 250

−1 − 9 e− t
4

.

This solution tends to the equilibrium solution P = 250 as t → ∞ and the solution goes to the other equilibrium
position as t → −∞. One can show that any solution with initial value satisfying 0 < P(0) < 250 has these prop-
erties. Thus in the case of the model involving the wolves the equilibrium solution P = 250 is stable, and the
equilibrium solution P = 0 is unstable. Consequently, if the wolves have been living in the region for a number
of years you would expect to find around 250 wolves living there.

In the qualitative study of mathematical models that use differential equations the equilibrium solutions are
important in that they are the solutions to which the system seems to tend to or to tend away from with increasing
time.

Let x′ = f (x) be a first order differential which does not involve the independent variable
t explicitly. Let f (c) = 0. We say that x = c is an equilibrium point.

• An equilibrium solution is a constant solution, φ(t) = c, where f (c) = 0.

• An equilibrium solution is stable if a small change in the initial conditions gives a
solution which approaches the equilibrium point as t → ∞.

• An equilibrium solution is unstable if a small change in initial conditions gives a so-
lution curve that moves away from from the equilibrium point as t → ∞.

Exercises 9.5

1. A certain radioactive substance has a half-life of 1740 years.

(a) Write a differential equation describing the decay of a sample of this substance and plot a graph of a
solution. From your graph of sample mass versus time, estimate the time required for the sample to
decrease to 25

(b) Compute this time from a solution formula for the sample size and compare with your result in (b).

2. At time t = 0 a parachutist who weighs 165 lbs opens the parachute at the height of 3000 ft when the velocity
is 88 ft/sec. The force of air resistance is given by 60v(t) lbs/sec, where v(t) the velocity of the parachutist
at time t.

(a) Write out a differential equation and initial value problem for v(t), and another differential equation
and initial value problem for x(t), the height of the parachutist above the ground t.

(b) Create a graph of v versus t, 0 ≤ t ≤ 1 min. What value does v(20) have?

(c) When will the parachutist hit the ground?

(d) What will the parachusist’s velocity be just before hitting the ground?

3. An aquarium can support no more than 225 tropical fish of a certain species. Nine of these fish are placed
into the aquarium. Assume that the rate of growth P′ of the fish is directly proportional to the population P
and the limiting factor 225 − P at any time t in weeks with proportionality constant r = 0.00225.

(a) Determine the fish population P(t) as an explicit function of time t.

(b) Make a Maple V plot of P(t).

(c) What values of P and t make sense in the problem situation.

(d) Make a direction field of the problem situation.

(e) When will the fish population be 100? 150?

(f) When will the fish population essentially reach the aquarium’s capacity?

9.6 Systems of Differential Equations

In the previous sections of this chapter it was shown how to analyze a single differential equation. Sometimes
an exact solution can be found explicitly, but in many cases one can only approximate the solutions numerically.
Even in the latter case it is possible to determine many of the salient features of the solutions by studying their
direction fields. Analyzing the behavior of solutions near equilibrium points and determining their stability, gives
much insight into the long term qualitative behavior of solutions. You should now be aware, from the examples of
mathematical models you have seen, that understanding the behavior of solutions to certain differential equations
leads to predictions about solutions of “real-world” problems. In this section we study the behavior of solutions
of systems of more than one differential equation.

Definition
Let f (t, x, y) and g(t, x, y) be functions of three variables t, x and y. A system of equations
of the form

x′ = f (t, x, y), y′ = g(t, x, y)

is called a first order system of differential equations. The variables x, and y are the de-
pendent variables and the variable t is called the independent variable. Differentiable func-
tions φ(t) and ψ which are defined on some interval I and such that the equations

φ′(t) = f (t, φ(t), ψ(t)), ψ′(t) = g(t, φ(t), ψ(t))

are satisfied for all t ∈ I is called a solution of the system of differential equations.

For example, the functions φ(t) = sin t, ψ(t) = cos t provide a solution for the system of differential equations

x′ = y, y′ = −x,

as can be seen by direct substitution.

Example 9.6.1 Analyze the solutions of the system of differential equations

x′ = −xy, y′ = xy − y.

Solution: Maple V will be used to study the solutions of this system numerically and geometrically.

> deq := diff(x(t),t)=-x(t)*y(t),diff(y(t),t)=x(t)*y(t)-y(t);

deq := d
dt

x(t) = −x(t)y(t),
d
dt

y(t) = x(t)y(t) − y(t)

As an illustration one can obtain a numerical solution of the system using dsolve with the numeric option satisfying
the initial value problem

x(0) = 2.5, y(0) = 0.1.

> sol := dsolve({deq,x(0)=2.5,y(0)=0.1},{x(t),y(t)},type=numeric);

sol := proc(rkf45_x) ... end

> sol(0); sol(2); sol(4);

[t = 0, x(t) = 2.500000000000000, y(t) = .1000000000000000]

[t = 2, x(t) = 1.174952482169065, y(t) = .6699844908362961]

[t = 4, x(t) = .3853721657880563, y(t) = .3447913533442888]

The above segment solves the initial value problem. The last maple V output shows that if (φ(t), ψ(t)) denotes
the solution then for values of t = 2 and t = 4 we have

φ(2) = 1.174952482169065, ψ(2) = .6699844908362961

and
φ(4) = .3853721657880563, ψ(4) = .3447913533442888.

This solution can be used to create plots, but rather, we will use Plot2 from DEtools to make plots.
It is often informative to find the points in the (x, y) plane where the direction field is parallel to the x-axes,

i.e., when y′ = 0, or is parallel to the y-axes i.e., when x′ = 0. Observe that x′ = 0 when −xy = 0 and y′ = 0 when
xy − y = 0. These curves are called the nullclines for the system. The next Maple V segment illustrates how to
draw these nullclines. Note that x′ = 0 on each coordinate axes, and y′ = 0 when y = 0 and x = 1. See Figure 65

> eq := -x*y=0,x*y-y=0;

eq := - x y = 0, x y - y = 0

> plt := plots[implicitplot]({eq},x=-1/2..3,y=-1..3):";

We now prepare to plot the direction field for the system, by using procedures from the DEtools package. Issue
the with command. See Figure 66.

> with(DEtools:

> plt1 := DEplot2([deq],[x,y],t=0..1,x=-1/2..3,y=-1/2..3):";

Figure 65: Nullclines for x′ = −xy, y′ = xy − y

Figure 66: Direction Field of x’=-xy,y’=xy-y

Figure 67: Nullclines for the system

A plot of the nullclines and direction field on the same graph is given in Figure 67. Observe that the direction
field along the nullclines is parallel to one of the coordinate axis.

> plots[display]({plt,plt1});

The procedure DEplot2 will now be used to obtain the solution curve in three dimensional space. Control of which
variables are plotted is achieved by assigning the scene option. In Figure 68 a plot of the solution curve, satisfying
the initial value problem x(0) = 2.5, y(0) = 0.1, in (t, x, y) space is obtained by using scene = [t,x,y].

> DEplot2([deq],[x,y],0..4,{[x(0)=2.5,y(0)=0.1]},x=-1/2..3,

> y=-1/2..3,scene = [t,x,y],axes = normal);

In Figure 69 DEplot2 with scene =[x,y] has been used to plot the solution curve that is shown in Figure 68 in
the (x, y)-plane along with the direction field.

> DEplot2([deq],[x,y],0..10,{[x(0)=2.5,y(0)=0.1]},x=-1/2..3,y=-1/2..3,

> scene = [x,y]);

Figure 68: Solution for x′ = −xy, y′ = xy − y in
(t, x, y)-space

Figure 69: Direction field along with solution

When you have a solution to a system of equations with x = φ(t), and y = ψ(t) you can also make plots of
each of these curves with DEplot2 by using scene = [t,x] and scene = [t,y] respectively. See Figures 70 and 71.

> DEplot2([deq],[x,y],0..10,{[x(0)=2.5,y(0)=0.1]},x=-1/2..3,y=-1/2..3,

> scene = [t,x]);

> DEplot2([deq],[x,y],0..10,{[x(0)=2.5,y(0)=0.1]},x=-1/2..3,y=-1/2..3,

> scene = [t,y]);

Systems Resulting From Mathematical Models

In the preceding section we studied population models that involved one species in isolation. Such models lead to
a single first order differential equation. When we have the ability to use more than one differential equation at a
time we can introduce models that involve more than one species.

Example 9.6.2 In this example we examine a system of differential equations that are derived from what is known
as predator-prey interaction. Let x(t), and y(t) denote the population of a predator species and a prey species
respectively. The predator-prey model assumes that (x(t), y(t)) satisfy the system:

x′ = (−a + by)x, y′ = (c − dx)y,

where a and c are positive numbers that are the decay (or death) and growth coefficients of each in the absence of
the other species. It is assumed that the number of predator-prey encounters is proportional to the population of
each. Thus b and d measure, respectively, predator efficiency in converting food (the prey) into fertility and the



              

9 ORDINARY DIFFERENTIAL EQUATIONS 112

-0.5

0

0.5

1

1.5

2

2.5

3

x

0 2 4 6 8 10
t

Figure 70: Solution for x′ = −xy, y′ = xy − y in (t, x)-
space

-0.5

0

0.5

1

1.5

2

2.5

3

y

0 2 4 6 8 10
t

Figure 71: Solution for x′ = −xy, y′ = xy − y in (t, y)-
space

the probability that an encounter removes one of the prey. Analyze the solution space for values of the parameter
given by a = 1, b = 1

100 , c = 2, and d = 2
25 .

Solution: First enter the equation into a Maple V session.

> deq := diff(x(t),t)=(-1+y(t)/100)*x(t),

> diff(y(t),t)=(2-2*x(t)/25)*y(t);

deq := d
dt

x(t) =
(

−1 + y(t)
100

)
x(t),

d
dt

y(t) =
(

2 − 2 x(t)
25

)
y(t)

Note that if there is no prey, i.e., that y(t) equal zero, then the system reduces to a single differential equation

x′ = −x

and the predators die off exponentially, since there is no food (the prey). Whereas, if there are no predators then
the single equation is

y′ = 2y,

and the prey explode exponentially and will soon exhaust their food supply.
It is usually productive to find the equilibrium points,

> equilibrii := solve({rhs(deq[1]),rhs(deq[2])},{x(t),y(t)});

equilibrii := {y(t) = 0, x(t) = 0}, {y(t) = 100, x(t) = 25}

The last Maple V output tells us that there are two equilibria points. An equilibrium (0, 0) means that if there
are no predators and no prey at a given time then there never will be. The equilibrium at (25, 100) means that if it
ever happens that there are 25 predators and 100 prey, then there will aways be that number. It is more interesting
to look at other solutions. Some plots using DEplot2 of solution curves in phase space ((x, y)- space), (t,x) space,
and (t,y) space for initial conditions

x(0) = 8, y(0) = 100; x(0) = 12, y(0) = 100; and x(0) = 18, y(0) = 100,

are shown in Figures 72,73, and 74.

Figure 72: Phase space for x′ = (−1 + y/100)x, y′ = (2 − 2x/25)

Figure 73: Solutions for x′ = (−1 + y/100)x, y′ =
(2 − 2x/25) in (t, x)-space

Figure 74: Solutions for x′ = (−1 + y/100)x, y′ =
(2 − 2x/25) in (t, y)-space

> with(DEtools):
> DEplot2([deq],[x,y],t=0..20,{[x(0)=8,y(0)=100],[x(0)=12,

> y(0)=100],[x(0)=18,y(0)=100]},arrows = NONE,x=0..60,y=0..300,

> scene = [x,y],stepsize = 0.1);

> DEplot2([deq],[x,y],t=0..20,{[x(0)=8,y(0)=100],[x(0)=12,

> y(0)=100],[x(0)=18,y(0)=100]},arrows = NONE,x=0..60,y=0..300,

> scene = [t,x],stepsize = 0.1);

> DEplot2([deq],[x,y],t=0..20,{[x(0)=8,y(0)=100],[x(0)=12,

> y(0)=100],[x(0)=18,y(0)=100]},arrows = NONE,x=0..60,y=0..300,

> scene = [t,y],stepsize = 0.1);

Populations that rise and fall as in this example exhibit a form of balance in the sense that both survive.
Example 9.6.3 A simple pendulum consists of a bob of mass m hanging on a (assumed to be massless) rigid

rod of fixed length L firmly attached to a horizontal support. The pendulum is in equilibrium when the bob and
rod are aligned with the local vertical and at rest. let x(t) denote the angle that the rod makes with the vertical, let
y(t) = x′(t) be angular velocity. Then it can be shown that (x(t), y(t)) satisfies the equations

x′ = y, y′ = − g
L

sin x − c
m

y,

where g is the acceleration due to gravity, and c is a constant due to friction. Suppose that g
L = 1, and c

m = 0.2,

plot the phase plane with the direction field, along with solution curves that have initial values

x(0) = 0, y(0) = 2; x(0) = 0, y(0) = 2.5; x(0) = 0, y(0) = 3; and x(0) = 0, y(0) = 3.5.

Solution: Enter the differential equation and the initial values into a Maple V session. Then apply DEplot2.

> deq := diff(x(t),t)=y(t),diff(y(t),t)=-sin(x(t))-0.2*y(t);

deq := d
dt

x(t) = y(t),
d
dt

y(t) = − sin(x(t)) − 0.2y(t)

The Maple V command that creates the direction field along with the solution curves satisfying the given initial
conditions in the (x, y)- plane is shown in Figure 75 and is written below.

> inits := {[x(0)=0,y(0)=2],[x(0)=0,y(0)=2.5],[x(0)=0,y(0)=3],

> [x(0)=0,y(0)=3.5]};
inits := {[x(0) = 0, y(0) = 2], [x(0) = 0, y(0) = 2.5],

[x(0) = 0, y(0) = 3], [x(0) = 0, y(0) = 3.5]}
> DEplot2([deq],[x,y],-10..20,inits,x=-10..10,y=-6..6,scene =[x,y],

> stepsize = 0.1);

If you wish to see the behavior of the same solution curves plotted with x vs t, then the next maple V command
creates Figure 76.

> DEplot2([deq],[x,y],-10..20,inits,x=-10..10,scene =[t,x],

> stepsize = 0.1);

As was mentioned in the statement of this example the pendulum has two equilbria: when the bob and rod are
aligned with the vertical and at rest. With the coordinates used here this means the points with coordinates (0, 0)

and (π, 0). These represent points such that x = 0, which means the rod and bob are hanging straight down and
are not moving, and the point with x = π, which means the rod and bob are balanced pointing straight up and are
not moving. Observe that the right hand sides of the differential equation vanish simultaneously at infinitely many
points: the points of the form (nπ, 0), for all integers n. You might think that this suggests that there are infinitely
many equilibria. Indeed the differential equation does have infinitely many equilibria, but only two in the context
of the equations as a model for a pendulum. Observe that the pendulum has the same position at the point (0, y)

as with any of the points (2nπ, y). In general the pendulum is in the same state when its coordinates are (x, y) or

Figure 75: Solutions for pendulum in (x, y)-space

Figure 76: Solutions for pendulum in (t, x)-space

(x + 2nπ, y). Thus the only two distinct equilibria occur when the state of the pendulum is given by any pair of
the form (2nπ, 0) ar (nπ, 0). The stability of these two points should be quite different, since a damped pendulum
which is swinging back and forth might be expected to eventually settle down in the position in which it is hanging
straight down. Moreover, if you try to balance a rod and bob straight up, the slightest push should send the rod and
bob into motion which will ultimately come to rest hanging straight down. Figure 76 shows the x-component of
three solutions for which initially x = 0 and the pendulum is set in motion at three different velocities: 2, 2.5, 3,
and 3.5. The first solution spirals to the point (0, 0) and from the graph it appears that |x(t)| remains less than π,

i.e., the pendulum does not make a full revolution. But the other three orbits tend to the point (2π, 0). This means
that make one complete revolution before settling down to the stable equilbrium state of hanging straight down.
The identification of points whose x-coordinates differ by integral multiples of 2π suggests that the phase space
for the pendulum is actually a cylinder rather than the plane.

Exercises 9.6

1. In the following locate all equilbria and the use Maple V to make a direction field plot that includes all of
the equlibria. In the vicinity of each equilibrium point fill in enough solution curves to determine whether
solutions approach the equibrium or not.

(a) x′ = 3x − 2y, y′ = 2x − 2y

(b) x′ = 4x − 2y, y′ = 8x − 4y

(c) x′ = −2x − y + 1, y′ = y − 1

(d) x′ = y2 − x2, y′ = y − 2x

(e) x′ = y2 + x2 − 4, y′ = y2 − x2

2. Red-Tail Hawks prey on the squirrel population on a certain college campus. Suppose that the number of
squirrels, x, and the number of Red-Tailed Hawks, y, are governed by the equations

x′ = 3x − xy, y′ = −125y − 3xy.

(a) Find all equilbrium points.

(b) Plot the direction field for the (x, y)-plane for part of the first quadrant. Include several solution curves.

(c) Plot the solution curves found in (b) in the (t, x) and (t, y) plane.

(d) Discuss how the two populations can be expected to vary with time.

3. Consider the system
x′ = −x + ay, y′ = −x − y.

Make graphs that show how the character of the direction fields, and solution curves change as a varies from
-1 to 1. For which values of a is there a sudden change in the nature of the solutions?

4. Consider an undamped pendulum
x′ = y, y = − sin(x).

(a) Plot (in the (x, y)-plane) the solution curves which which are initially at the points:

(−12, 1), (−12, 1.5), (−12, 2), and (−12, 3).

Explain the different kind of motions of the pendulum correspond to closed and nonclosed curves.

(b) Repeat part (a) with solutions that originate at

(−6, b), (0, b), and (6, b),

for some value of b.

(c) Consider the closed curves which correspond to periodic motions of the pendulum originating at

(1, 0), (1.5, 0), (2, 0), and (3, 0).

Plot x versus t and estimate the period T of each solution. How does T depend on the initial position?

(d) It can be shown that the solution which has initial value x(0) = α, y(0) = 0 has a period equal to

T = 4
∫ π/2

0

dτ√
1 − k2 sin2 τ

, where k = sin(
α

2
).

Evaluate this integral numerically when α = 1, 1.5, 2, and 3. Compare your answer with the results
of part (c).

9.7 Second-Order Linear Differential Equations

Let a 6= 0, and b be real numbers. A differential equation of the form

ax′ + bx = 0

is called a first order linear differential equation with constant coefficients. It is an equation that can be solved
either by the technique of separation of variables or by inspection. The general solution is

x(t) = Ce− b
a t.

Observe that the coefficient of t in the exponent is a root of the first degree polynomial equation in λ,

aλ + b = 0.

The latter polynomial equation is called the characteristic equation for the differential equation

ax′ + bx = 0.

Now let a 6= 0, b and c be real numbers. A differential equation of the form

ax′′ + bx′ + cx = 0

is called second order linear differential equation with constant coefficients. It will now be shown how to find the
general solution to this equation. The solution should be a function which when added to a linear combination of
its first and second derivative gives zero. Thus a solution of the form

φ(t) = eλt

where λ is to be determined is sought. Substituting this function for x into the differential equation gives

a(λ2eλt) + b(λeλt) + ceλt = 0.

Factoring eλt from the left-hand side leads to the equation

(aλ2 + bλ + c)eλt = 0,

which must be satisfied for all t. Recalling that eλt can never be zero and thus can be divided out of the above
equation leads to the second degree polynomial equation in λ,

aλ2 + bλ + c = 0.

This equation is called the characteristic equation for the differential equation. If λ is a root of the characteristic
equation then x = eλt is a solution of the differential equation. The characteristic equation is easy to obtain: all
one does is replace x by 1, x′ by λ, and x′′ by λ2 in the differential equation.

Maple V can be used to make these calculations. Enter the differential equation:

> deq := a*diff(x(t),t$2)+b*diff(x(t),t)+c*x(t)=0;

deq := a
d2

dt2
x(t) + b

d
dt

x(t) + cx(t) = 0

Substitute x(t) = eλt into this equation.

> e1 := subs(x(t)=exp(lambda*t),deq);

e1 := a
d2

dt2
eλ t + b

d
dt

eλ t + ceλ t = 0

> e2 := simplify(e1);

e2 := aλ2eλ t + bλ eλ t + ceλ t = 0

Now divide both sides by eλt to obtain the characteristic equation. Note the usage of expand and simplify.

> ceq := simplify(expand(e2/exp(lambda*t)));

ceq := aλ2 + bλ + c = 0

Each root to the characteristic equation leads to a solution of the differential equation. The following fact shows
a way to obtain more solutions.

Let φ1(t) and φ2(t) be two solutions of the second order linear equation

ax′′ + bx′ + cx = 0

then if C1, and C2 are numbers

φ(t) = C1φ1(t) + C2φ2(t)

is also a solution

This fact can be verified by direct substitution: Substitute C1φ1(t) + C2φ2(t) for x in the differential equation
and rearrange the terms to obtain

a(C1φ1 + C2φ2)
′′ + b(C1φ1 + C2φ2)

′ + c(C1φ1 + C2φ2) =
C1(aφ′′

1 + bφ′
1 + cφ1) + C2(aφ′′

2 + bφ′
2 + cφ2) = 0 + 0 = 0.

The next fact gives a condition which guarantees when two solutions of the equation can be used to generate so-
lutions of all initial value problems.

Let φ1(t) and φ2(t) be two solutions of the second order linear equation

ax′′ + bx′ + cx = 0.

Suppose that for some t0 the inequality is true

1 = φ1(t0)φ
′
2(t0) − φ′

1(t0)φ2(t0) 6= 0.

Let x0, and x′
0 be numbers. Then there are numbers C1 and C2 such that the function

φ(t) = C1φ1(t) + C2φ2(t)

is a not only a solution of the differential equation but also satisfies the initial conditions
φ(t0) = x0, φ′(t0) = x′

0.

Since φ(t) = C1φ1(t)+ C2φ2(t) is a solution for every pair of numbers C1 and C2, it follows that the statement
in the box will be true if C1 and C2 can be found so that

φ(t0) = C1φ1(t0) + C2φ2(t0) = x0, φ′(t0) = C1φ
′
1(t0) + C2φ

′
2(t0) = x′

0.

Recognizing that these two algebraic equations are linear in C1 and C2 enables one to write the solution

C1 = x0φ
′
2(t0) − x′

0φ2(t0)

1
, C2 = x′

0φ1(t0) − x0φ
′
1(t0)

1
.

Example 9.7.1 Find the general solution to the second order differential equation

x′′ − 3x′ + 2x = 0.

Plot a few solution curves.

Solution: The characteristic equation is
λ2 − 3λ + 2 = 0.

By factoring the right hand side it follow that the last equation is

(λ − 1)(λ − 2) = 0.

This means that two solutions are φ1(t) = et, and φ2(t) = e3t. This gives the general solution since

1 = φ1(t0)φ
′
2(t0) − φ′

1(t0)φ2(t0) = et0 (2e2t0 ) − et0 e2t0 = e3t0 6= 0,

and we can solve every possible initial condition. We now will plot the solutions which satisfy the three sets of
initial conditions

x(0) = −1, x′(0) = −1; x(0.5) = 0, x′(0.5) = 1; x(−0.5) = 0, x′(−0.5) = 1.

The general solution can be defined to Maple V as follows.

> gensol := C1*exp(t)+C2*exp(2*t);

gensol := C1 et + C2 e2 t

The constants C1, C2 are solved for the initial condition x(0) = −1, x′(0) = −1 by the following statement.

> Con1 := solve({subs(t=0,gensol)=-1,subs(t=0,
> diff(gensol,t))=-1},{C1,C2});

Con1 := {C2 = 0,C1 = −1}
Substituting this result into the general solution gives the particular solution that satisfies the initial value problem.

> x1 := simplify(subs(Con1,gensol));

x1 := −et

The same Maple V steps can be followed to obtain the other two solutions.

> Con2 := solve({subs(t=0.5,gensol)=0,subs(t=0.5,diff(gensol,t))=1},{C1,C2});

Con2 := {C2 = .3678794412, C1 = -.6065306596}
> x2 := simplify(subs(Con2,gensol));

x2 := {C2 = 0,C1 = −1}

> Con3 := solve({subs(t=-0.5,gensol)=0,subs(t=-
> 0.5,diff(gensol,t))=1},{C1,C2});

Con3 := {C2 = 2.718281828, C1 = -1.648721271}

> x3 := simplify(subs(Con3,gensol));

x3 := −1.648721271et + 2.718281828e2.0t

The plot of all three solutions is given in Figure 77

> plot({x1,x2,x3},t=-1..1,x=-4..4);

Figure 77: Solution curves for x′′ − 2x′ + 2x = 0

You should learn to solve simple second order linear differential with constant coefficients by hand. Probably
there will be problems in which the computation becomes burdensome and thus you should also learn how to solve
these equations exactly using Maple V. There follows a Maple V segment that indicates the relevent commands.

> deq := diff(x(t),t$2)-3*diff(x(t),t)+2*x(t)=0;

deq := d2

dt2
x(t) − 3

d
dt

x(t) + 2 x(t) = 0

The next step uses dsolve to find the general solution.

> gensol2 := dsolve(deq,x(t));

gensol2 := x(t) = C1 et + C2 e2 t

If you have a general solution, then you can use it to find the constants using solve from the initial conditions.

> constants := solve({subs(t=0,rhs(gensol2))= -1,subs(t=0,

> diff(rhs(gensol2),t))=-1},{_C1,_C2});

constants := { C1 = −1, C2 = 0}
Now the solution to the initial value problem is

> xx1 := subs(constants,rhs(gensol2));

xx1 := −et

This agrees with the answer which was obtained previously. You can also use dsolve to find the solution to the
other initial value problem.

> xx2 := dsolve({deq,x(0.5)=0,D(x)(0.5)=1},x(t));

xx2 := x(t) = −0.6065306596et + 0.3678794412e2 t

This also agrees with the previously found solution. Note that the Maple V output is in the form of an equation,
and thus you may need to use rhs when preparing to work with the solution. The solution to the third initial value
problem can be found in the same way and is left as an exercise.

The Characteristic Equation

Since the characteristic equation
aλ2 + bλ + c = 0

is so important to solving the differential equation

ax′′ + bx′ + cx = 0,

it will now be analyzed. The graph of the second degree polynomial

y = aλ2 + bλ + c

is parabola which is concave up or concave down depending on the sign of a. There are thus three possibilities for
the roots of the equation. See Figure 78 for the three possibilties that can occur when a > 0. The graphs for a < 0
are similar except are concave down.

Inspection of Figure 78 indicates that the roots of the characteristic will be one of the following:

1. two distinct real roots, r1, and r2 (when the vertex of the parabola is below the λ-axis),

2. one double root, r, (when the vertex of the parabola is tangent to the λ-axis,

3. a pair of complex conjugate roots (when the vertex of the parabola is above the λ-axis.

To see this algebraically we simply solve the quadratic equation

aλ2 + bλ + c = 0.

Figure 78: Possible positions of y = aλ2 + bλ + c

The quadratic formula gives

λ = b ±
√

b2 − 4ac
2a

.

Accordingly the roots will be real and distinct if the discriminant satisfies

b2 − 4ac > 0.

Let

r1 = b −
√

b2 − 4ac
2a

, r2 = b +
√

b2 − 4ac
2a

.

The functions φ1(t) = er1t, and φ2(t) = er2t are solutions. Observes that

φ1(t0)φ
′
2(t0) − φ′

1(t0)φ2(t0) = (r2 − r1)e
r1t0+r2t0 6= 0,

and hence
φ(t) = C1er1t + C2er2t

is a general solution.
In the second case when the roots merge into a single root the discriminant must vanish

b2 − 4ac = 0

and hence

r = − b
2a

.

In this case we only get one solution, φ1(t) = ert, by the procedure that is under discussion. It turns out that an-
other solution is given by φ2(t) = tert. This follows by direct substitution after recognizing that in this case the
characteristic equation has the form

a(λ2 − 2rλ + r2) = 0

which implies that the differential equation is

a(x′′ − 2rx′ + r2x) = 0.

Thus the general solution is
φ(t) = (C1 + C2t)ert,

since
φ1(t0)φ

′
2(t0) − φ′

1(t0)φ2(t0) = ert0 6= 0.

Finally, when the discriminant satisfies
b2 − 4ac < 0,

the characteristic equation has complex conjugate roots

r1 = b − i
√

4ac − b2

2a
, r2 = b + i

√
4ac − b2

2a
,

which, for purposes of simplification, will be written as

r1 = α − iβ, r2 = α + iβ.

Two solutions are then given by

ψ1(t) = e(α−iβ)t) = eαte−iβt, ψt(t) = e(α+iβ)t) = eαteiβt.

These solutions are not so pleasing, since they are complex valued, and, in this course, we are interested only in
real-valued solutions. Recall that

cos βt = eβt + e−βt

2
, and sin βt = eβt − e−βt

2i
.

This means that the functions defined by

φ1(t) = ψ1(t) + ψ2(t)
2

= eαt cos βt, φ2(t) = ψ1(t) − ψ2(t)
−2i

= eαt sin βt

are solutions and are real valued. Furthermore,

φ1(t0)φ
′
2(t0) − φ′

1(t0)φ2(t0) = βe2αt0 6= 0.

Thus a general solution for this case is

φ(t) = eαt(C1 cos βt + C2 sin βt).

Oscillations

Suppose a mass m is attached to the end of a (massless) spring, the other end of which is attached to a solid hori-
zontal beam. A coordinate system is established along the spring’s axis and when the spring-mass configuration is
in equilibrium the coordinate for the mass is zero and measures positive in the downward direction. It is assumed
that the restoring force for the spring obeys Hooke’s Law which means that if the mass is displaced to the point
with coordinate x then the force is given by

F = kx,

for a contant k. In the absence of damping Newton’s Second Law implies that the position x(t) of the mass satisfies
the differential equation

mx′′ = −kx.

Sometimes it is assumed that the spring has damping which is proportional to the velocity of the mass and acts in
a direction which is opposite to the motion. In this case x(t) satisfies the differential equation

mx′′ = −cx′ − kx.

Example 9.7.2 Let the differential equation for a certain spring-mass system be

x′′ + 4x = 0.

1. The mass is released from rest at a distance 2 units below the equilibrium position. Find a formula that gives
the position of the mass as a function of time.

2. The mass is set in motion with a velocity of -3 ft/sec from a point a distance 2 units below the equilibrium
position. Find a formula that gives the position of the mass as a function of time.

3. Express each of the solutions in the form

x(t) = A cos(ωt + θ).

4. Make a Maple V plot of both solutions on the same graph over a time period equal to twice the period of the
solutions.

Solution: The characteristic equation for the differential equation is

λ2 + 4 = 0.

The characteristic roots are ±2i. In terms of the discussion above this means that the roots of the characterisic
equation are complex with α = 0, and β = 2. More precisely, one says that the roots are pure imaginary. The
general solution is

φ(t) = C1 cos 2t + C + 2 sin 2t.

If the mass is released from rest at a point 2 ft below the equilibrium position the solution must satisfy the initial
conditions

φ(0) = 2, φ′(t) = 0.

The equations that determine C1 and C2 are

φ(0) = C1 cos(0) + C2 sin(0) = C1 = 2

φ′(0) = −2C1 sin(0) + C2 cos(0) = C2 = 0.

The solution is
φ1(t) = 2 cos(2t).

The solution in the second problem satisfies

φ(0) = C1 cos(0) + C2 sin(0) = C1 = 2

φ(0) = −2C1 sin(0) + C2 cos(0) = C2 = −3.

The second solution is
φ2(t) = 2 cos(2t) − 3 sin(2t).

The first solution is already in the form
x(t) = A cos(ωt + θ),

with A = 2 and θ = 0. Recall the addition formula for cos,

cos(A + B) = cos A cos B − sin A sin B.

This means the second solution can be written as

φ2(t) = 2 cos(2t) − 3 sin(2t) =
√

13(cos(2t)
2√
13

− sin(2t)
3√
13

) =
√

13 cos(2t + θ),

where tan(θ) = sin θ
cos θ

= 3/2. Thus θ = arctan(3/2) which is approximately equal to

> theta := evalf(arctan(3/2));

θ := 0.9827937232

To check the accuracy the following can be used to check the result:

> evalf(sqrt(13)*cos(theta)); evalf(sqrt(13)*sin(theta));

2.000000000

2.999999999

Thus the answer checks to 8 decimal places. The example will be completed by giving a plot of the two solutions.
The period of each solution is π, so the plot is over a time length of 2π. See Figure 79,

> plot({2*cos(2*t),2*cos(2*t)-3*sin(2*t)},t=0..2*Pi);

Figure 79: Two solutions of x′′ + 4x = 0.

Example 9.7.3 Let the differential equation for a certain spring-mass system with damping is

x′′ + 0.1x′ + 0.2x = 0.

1. The mass is released from rest at a distance 2 units below the equilibrium position. Find a formula that gives
the position of the mass as a function of time.

2. The mass is set in motion with a velocity of -3 ft/sec from a point a distance 2 units below the equilibrium
position. Find a formula that gives the position of the mass as a function of time.

3. Express each of the solutions in the form

x(t) = Aeαt cos(ωt + θ).

4. Make a Maple V plot of both solutions on the same graph over a time interval [0, 30].

Solution: The characteristic equation for the differential equation is

λ2 + 0.1 λ + 0.2.

Maple V will ease the burden of some the calculations.

> eq := lambda^2+0.1*lambda+0.2=0;

eq := λ2 + 0.1λ + 0.2 = 0

The characteristic roots can be found using solve.

> sol := solve(eq,lambda);

sol := - .05000000000 + .4444097209 I,
- .05000000000 -.4444097209 I

These roots are complex conjugate with α = −0.05 and β = .4444097209. This means that the general solution is

φ(t) = e.05t(C1 cos(.4444097209t) + C2 sin(.4444097209t).

Note that α is the real part and β is the imaginary part of the complex number α + iβ. This suggests the use
of the Maple V commands Re and Im to find the general solution.

> phi := exp(Re(sol[1])*t)*(C1*cos(Im(sol[1])*t)+C2*sin(Im(sol[1])*t));

phi := e−0.05000000000t (C1 cos(0.4444097209t) + C2 sin(0.4444097209t))

The constants are found for the first initial value problem.

> cons1 := solve({subs(t=0,phi)=2,subs(t=0,diff(phi,t))=0},{C1,C2});

cons1 := {C1 = 2., C2 = .2250175802}

The solution to the first initial value problem is obtained by substituting these values for the C’s.

> xx1 := subs(cons1,phi);

xx1 := e−0.05000000000t (2.0 cos(0.4444097209t) + 0.2250175802 sin(0.4444097209t))

The last Maple V output is the desired solution. The solution to the second initial value problem follows similarly.

> cons2 := solve({subs(t=0,phi)=2,subs(t=0,diff(phi,t))=-3},{C1,C2});

cons2 := {C1 = 2., C2 = -6.525509825}

> xx2 := subs(cons2,phi);

xx2 := e−0.05000000000t (2.0 cos(0.4444097209t) − 6.525509825 sin(0.4444097209t))

The next thing to do is to express the solution in the form

Aeαt cos(ωt + θ).

The next Maple V segment finds A and θ for the first solution.

> A1 := subs(cons1,sqrt(C1^2+C2^2));

A1 := 2.012618422

> tan1 := subs(cons1,-C2/C1);

tan1 := -.1125087901
> theta1 := arctan(tan1);

θ1 := −0.1120376427

Thus the first solution can be written as

φ1(t) = 2.012618422e−.05t cos(0.4444097209t − 0.1120376427)

The second solution can be treated similarly.

> A2 := subs(cons2,sqrt(C1^2+C2^2));

A2 := 6.825121133

> tan2 := subs(cons2,-C2/C1);

tan2 := 3.262754913

> theta2 := arctan(tan2);

θ2 := 1.273396582

This means that the second solution can be written as

φ2 = 6.825121133e−.05t cos(0.4444097209t + 1.273396582)

The next two maple V commands represent checks of the correctness of the preceding calculations.

> expand(A1*cos(Im(sol[1])*t+theta1));
2.000000000 cos(.4444097209 t) + .2250175803 sin(.4444097209 t)

> expand(A2*cos(Im(sol[1])*t+theta2));

1.999999997 cos(.4444097209 t) - 6.525509826 sin(.4444097209 t)

It follows that, except for roundoff error, the above calculations are correct. See Figure 80 for the plot of the two
solutions.

> plot({xx1,xx2},t=0..30);

Figure 80: Two solutions of x′′ + 0.1x′ + 0.2x = 0.

Exercises 9.7

1. Find the characteristic equation and use it to obtain the general solution for each of the following second
order linear differential equations.

(a) x′′ − 3x′ − 10x = 0

(b) x′′ + 10x′ + 25 = 0

(c) x′′ + 4x′ + 13x = 0

2. In each of the following problems the motion of a mass attached a spring is described by an initial value
problem.

(a) Solve each of the problems.

(b) Plot the solution in the specified interval.

(c) Find the maximum of the solution in the specified interval.

(d) When possible express each solution in the form

Aeαt cos(ωt + θ).

(a) x′′ + 15 = 0, x(0) = −2, x′(0) = 5; 0 ≤ t ≤ 5

(b) x′′ + 14.9x = 0, x(0) = −2, x′(0) = 5; 0 ≤ t ≤ 5

(c) x′′ + 5.1x′ + 6x = 0, x(0) = −1.2, x′(0) = 3; 0 ≤ t ≤ 4

(d) x′′ + 3.9x′ + 18.73x = 0, x(0) = 3, x′(0) = 2; 0 ≤ t ≤ 4

3. In order to examine changes in the amplitude of an oscillation make Maple V plots of the functions

φ(t) = A cos(t + 1), for A = 0, 1/2, 1, 2,

on the same graph.

4. In order to examine how changes phase shift changes the graph make Maple V plots of the functions

φ(t) = cos(t + θ), for θ = 0, 1/2, 1, 2,

on the same graph.

5. In order to examine how changes phase shift changes the graph make Maple V plots of the functions

φ(t) = cos(ωt + 1), for ω = 0, 1/2, 1, 2,

on the same graph.

